



Sophia

Documentation

فهرست مطالب

۱.	مقدمه	۴
۲.	ساختار کلی	۴
۲-۱.	قواعد کلی نحو	۴
۲-۲.	کامنت‌ها	۴
۲-۳.	قواعد نامگذاری کلاس‌ها، متدها و متغیرها	۴
۳.	کلاس‌ها و متدها	۵
۳-۱.	ارث بری	۷
۳-۲.	کلیدواژه this	۸
۴.	متغیرها	۸
۴-۱.	متغیرهای محلی	۹
۵.	انواع داده	۹
۵-۱.	تایپ‌های پایه	۹
۵-۲.	اشاره‌گر تابع	۹
۵-۳.	لیست	۱۰
۶.	عملگرها	۱۲
۶-۱.	عملگرهای حسابی	۱۲
۶-۲.	عملگرهای مقایسه‌ای	۱۳
۶-۳.	عملگرهای منطقی	۱۳
۶-۴.	عملگر تخصیص	۱۴
۶-۵.	اولویت عملگرها	۱۴
۷.	ساختار تصمیم‌گیری	۱۵
۸.	ساختار تکرار	۱۶

۱۶.....	۸-۱ . حلقه‌ی for
۱۷.....	۸-۲ . حلقه‌ی foreach
۱۷.....	۸-۳ . کلیدواژه break
۱۷.....	۸-۴ . کلیدواژه continue
۱۷.....	۹ . scope ها
۱۷.....	۹-۱ . scope های موجود در زبان
۱۸.....	۹-۲ . قوانین scope ها
۱۸.....	۹-۳ . قوانین خطوط برنامه
۱۸.....	۱۰ . توابع پیش فرض
۱۹.....	۱۱ . نمونه کد

۱. مقدمه

زبان Sophia یک زبان شی گراست و برخی از ویژگی‌های زبان‌های شی گرا نظیر ارث‌بری را داراست.

در این زبان، یک کلاس اصلی وجود دارد که در آن یک متد constructor پیاده‌سازی شده‌است. برنامه‌هایی که در این زبان نوشته می‌شوند، در هنگام اجرا دستورات درون این متد را اجرا می‌کنند.

۲. ساختار کلی

در این زبان، کد برنامه درون یک فایل با پسوند .sop قرار دارد. این فایل شامل:

- یک یا چند کلاس
- یک کلاس اصلی (Main) شامل یک متد constructor
- هر کلاس شامل تعدادی متغیر و متد

۲-۱. قواعد کلی نحو

زبان Sophia به بزرگ و کوچک بودن حروف حساس است. در این زبان، وجود کاراکترهای tab و space تاثیری در خروجی برنامه ندارند. جزئیات مربوط به scope و خطوط برنامه در ادامه توضیح داده خواهد شد.

۲-۲. کامنت‌ها

در این زبان کامنت‌ها تک خطی هستند و تمامی کاراکترهای بعد از // تا انتهای خط کامنت به حساب می‌آیند.

۲-۳. قواعد نامگذاری کلاس‌ها، متدها و متغیرها

اسامی انتخابی برای نام‌گذاری باید از قواعد زیر پیروی کنند:

- تنها از کاراکترهای `A..Z`، `a..z`، `_` و ارقام تشکیل شده باشند (محدودیتی روی تعداد کاراکترهای یک اسم در زبان Sophia وجود ندارد).
- با رقم شروع نشوند.
- معادل کلیدواژه‌ها نباشند. در جدول زیر تمامی کلیدواژه‌های زبان Sophia آمده است:

class	extends	this	def	func	return
if	else	for	foreach	continue	break
false	true	int	boolean	string	void
list	in	null	new	print	

- نام هر کلاس یکتاست.
- نام هر متد در هر کلاس یکتاست. به عبارتی `method overloading` در این زبان وجود ندارد.
- نام هر متغیر در هر `scope` یکتاست ولی در `scope` های درونی تر از نام های متغیر بیرونی میتوان استفاده کرد. در نتیجه در طول آن `scope` متغیر درونی هنگام استفاده ارجحیت دارد.
- امکان تعریف دو متد و متغیر با نام یکسان در یک کلاس وجود ندارد.

۳. کلاس‌ها و متدها

در تعریف هر کلاس، در ابتدا نام آن و نام کلاسی که از آن ارث بری می‌کند (در صورت وجود) مشخص می‌گردد. سپس `scope` این کلاس توسط `{ }` مشخص می‌شود.

```

1 class A extends B {
2     // class body
3 }
```

متغیرها و متدهای هر کلاس می‌توانند با هر ترتیبی در `scope` آن کلاس تعریف شوند. مثال زیر نمونه‌ای از تعریف یک کلاس است:

```

1 class A {
2     x: int;
3     y: bool;
4     def A(p:int, q: bool) {
5         this.x = p;
6         this.y = q;
7     }
8     def void foo() {
9         print("foo");
10    }
11 }

```

کلاس Main، باید حتما یک constructor بدون آرگومان داشته باشد. کلاس Main فرزند هیچ کلاس دیگری نیست. کلاس‌ها می‌توانند یک یا چند متغیر و متد (تابع) داشته باشند. در هر متد نیز تعریف متغیرها در ابتدای آن و قبل از همه‌ی دستورات انجام می‌شود. همچنین در این زبان، همه متدها بجز constructor باید مقدار بازگشتی داشته باشند و در صورتی که مقدار بازگشتی متدها از نوع void نباشد، باید یک دستور return در متد وجود داشته باشد. آرگومان‌های متدها به صورت pass by value به متدها داده می‌شوند.

به عنوان مثال، تعریف یک متد constructor به صورت زیر انجام می‌شود:

```

1 def A(a: int, b:string) {
2     c: int;
3     c = a * 5 % 2;
4     this.a = c;
5     this.b = b;
6 }

```

نمونه ای از تعریف متدهای دیگر:

```
1 def int foo(n: int) {  
2     print("foo");  
3     return n + 1;  
4 }
```

برای فراخوانی متدها از () استفاده می‌کنیم. به مثال زیر توجه کنید:

```
1 x: int;  
2 x = this.foo(2);
```

برای ساختن یک نمونه^۱ جدید از یک کلاس، به صورت زیر از کلیدواژه new برای آن استفاده می‌کنیم:

```
1 a: A;  
2 a = new A(2, "Sophia");
```

۳-۱. ارث بری

هر کلاسی می‌تواند از حداکثر یک کلاس دیگر ارث بری کند (با استفاده از کلیدواژه extends) و از فیلدها و متدهای آن استفاده کند. قابلیت overriding برای متدهای یک کلاس وجود ندارد. همچنین overloading برای متغیرهای کلاس^۲ در این زبان وجود ندارد.

^۱ instance

^۲ برد. امکان تعریف متغیر همنام در یک کلاس و کلاسی که از آن به ارث می‌

۳-۲. کلیدواژه this

کلیدواژه this به کلاسی که در آن قرار داریم اشاره می‌کند و با استفاده از این کلیدواژه می‌توان به فیلدها و متدهای آن دسترسی پیدا کرد.

۴. متغیرها

تعریف متغیرها در زبان Sophia به صورت زیر انجام می‌شود:

```
1 identifier: type;
```

در این زبان نمی‌توان چندین متغیر را در یک خط تعریف کرد و یا در هنگام تعریف یک متغیر، آن را مقداردهی کرد.

در صورتی که به متغیری مقداری نسبت داده نشود، مقدار آن برابر با مقدار پیش فرض تایپ خود در نظر گرفته می‌شود. مقادیر پیش فرض تایپ‌های مختلف در جدول زیر آمده‌است:

int	0
bool	false
string	""

در صورتی که متغیر از جنس یک کلاس یا اشاره‌گر تابع باشد، مقدار اولیه آن null و در صورتی که لیست باشد، مقدار اولیه هرکدام از اعضای آن مطابق با مقدار اولیه تایپ آن عضو می‌شود.

۴-۱. متغیرهای محلی

تمامی متغیرهای محلی فقط در ابتدای متدها تعریف می‌شوند و در ابتدای هیچ scope جدید دیگری قابلیت تعریف متغیر محلی وجود ندارد.

یک متغیر در صورتی می‌تواند به جای دیگری استفاده شود (مثلا در آرگومان‌های توابع) که نوع آن زیرنوع^۱ متغیر دیگر باشد.

۵. انواع داده

۵-۱. تایپ‌های پایه

در زبان Sophia، سه تایپ پایه^۲ `int`، `string` و `bool` وجود دارند. در متغیرهای از نوع تایپ‌های پایه، خود مقادیر ذخیره می‌شود و نه اشاره‌گری به خانه‌ای از حافظه. علاوه بر آن‌ها، هر متغیر می‌تواند از جنس یکی از کلاس‌هایی باشد که در برنامه تعریف شده‌است. همچنین تایپ `void` نشانگر مقدار بازگشتی توابع بدون خروجی است. استفاده از این تایپ به عنوان عملوند و تعریف متغیر از این تایپ غیرمجاز است. تایپ‌های دیگر در ادامه آمده است.

۵-۲. اشاره‌گر تابع^۳

یک متغیر از نوع اشاره‌گر تابع، به یک تابع (متد) در برنامه اشاره می‌کند. برای مثال یک متغیر از جنس `<int, string->int` به یک تابع با ورودی‌های به ترتیب از جنس `int` و `string` و خروجی از نوع `int` اشاره می‌کند. برای توصیف تابعی بدون ورودی و خروجی از `<void->void` استفاده می‌کنیم. دقت کنید که توصیفاتی از قبیل `<void, int->string` نامعتبر هستند. برای فراخوانی تابعی که اشاره‌گر به آن اشاره می‌کند، از `()` استفاده می‌کنیم. به مثال زیر توجه کنید:

^۱ subtype

^۲ primitive

^۳ function pointer



```
1 x: int;  
2 fptr: func<int->int>;  
3 fptr = this.foo;  
4 x = fptr();
```

۵-۳. لیست

لیست مجموعه‌ی مرتبی با اندازه‌ی مشخص از انواع داده‌های مختلف است. در طی اجرای برنامه اندازه‌ی یک لیست ثابت باقی می‌ماند. برای مثال، `[1, "Sophia", [2, 3]]` نمایشگر یک لیست سه تایی است که عنصر آخر آن خود یک لیست دوتایی دیگر است. در حالت کلی لیست را به صورت لیستی از نوع داده‌های درون آن تعریف می‌کنیم. به عنوان مثال:



```
1 myList: list(int, func<int->int>, bool);  
2 myList = [2, this.foo, true];
```

در تعریف لیست می‌توان به بعضی یا تمام اعضای آن لیست شناسه نسبت داد:



```
1 myList: list(number: int, fptr: func<int->int>, isEven: bool);  
2 myList.number = 2;  
3 myList.fptr = this.foo;  
4 myList[2] = true;
```

در صورتی که تمامی اعضای یک لیست از یک نوع باشند، می‌توان آن را به صورت زیر تعریف کرد. در تعریف لیست‌ها به صورت زیر، طول لیست یک عدد ثابت است و استفاده از متغیرها در آن مجاز نیست. طول لیست نمی‌تواند صفر یا عددی منفی باشد.

```
1 myList2: list(2 # string);
2 myList2 = ["foo", "bar"];
```

برای دسترسی به عناصر لیست می‌توان از شناسه آنها (در صورت وجود) و یا اندیس آنها (که از ۰ شروع می‌شود) استفاده کرد. اندیس‌ها از جنس `int` هستند. بررسی مجاز بودن مقدار عددی اندیس و خارج از محدوده نبودن آن، به صورت `runtime` است. هنگام دسترسی به اعضای لیست، در صورتی که اعضای لیست نوع یکسانی نداشته باشند (لیست به صورت `list(length # type)` تعریف نشده باشد) اندیس‌ها باید عدد ثابت باشند.

نوع یک لیست زیرنوعی از نوع لیست دیگر است، اگر تعداد اعضای آن با لیست دیگر برابر و نوع اعضای آن زیرنوعی از نوع اعضای متناظر در لیست دیگر باشد. به مثال زیر توجه کنید که در آن کلاس `A` از کلاس `B` ارث‌بری کرده است:

```
1 myList1: list(radius:int, center: list(x:int, y:int), widget:B);
2 myList2: list(border:int, list(int, int), A);
3 myList3: list(int, list(int, int), int, A);
4 myList4: list(bool, list(int, int), string);
```

در این مثال، نوع `myList2` زیرنوعی از `myList1` است اما این موضوع برای `myList3` و `myList4` برقرار نیست.

در صورتی که یک لیست (لیست اول) زیرنوعی از لیست دیگر (لیست دوم) باشد، می‌توان آن را به جای لیست دوم استفاده کرد. در مثال زیر، در صورتی که یک تابع آرگومانی از نوع `myList1` داشته باشد، می‌توان یک متغیر از جنس `myList2` را به آن داد. اما در نهایت اسامی اعضای لیست همان اسامی تایپ نهایی (در اینجا `myList1`) خواهد بود. در این مثال هم کلاس `A` از کلاس `B` ارث‌بری کرده است.

```

1 myList1: list(radius:int, center: list(x:int, y:int), widget:B);
2 myList2: list(border:int, list(int, int), A);
3 myList1 = myList2;
4 print(myList1.radius); // valid reference
5 print(myList1.border); // invalid reference

```

اگر اعضای متناظر دو لیست با یکدیگر برابر باشند، آن دو لیست مساوی هستند.

۶. عملگرها

عملگرها در زبان Sophia به چهار دسته‌ی عملگرهای حسابی، مقایسه‌ای، منطقی و تخصیص تقسیم می‌شوند که شرح آنها در ادامه آمده‌است. توجه داشته باشید که عباراتی مانند $i++$ گزاره نیستند.

۶-۱. عملگرهای حسابی

این دسته از عملگرها تنها روی اعداد عمل می‌کنند. لیست این عملگرها در جدول زیر آمده است. در مثال‌های استفاده شده A را برابر با ۲۰ و B را برابر با ۱۰ در نظر بگیرید.

عملگر	شرکت‌پذیری	توضیح	مثال
+	چپ	جمع	$A + B = 30$
-	چپ	تفریق	$B - A = -10$
*	چپ	ضرب	$A * B = 200$
/	چپ	تقسیم	$A / B = 2$ $B / A = 0$
%	چپ	باقی‌مانده	$A \% B = 10$
-	راست	منفی‌تک‌عملوندی پیشوندی	$A = -20$

--A	پیشوندی	راست	-- و ++
A++	پسوندی	چپ	-- و ++

۶-۲. عملگرهای مقایسه‌ای

این عملگرها وظیفه‌ی مقایسه را دارند؛ پس نتیجه‌ی آنها باید مقدار true یا false باشد. یعنی خروجی آنها یک bool است.

توجه داشته باشید که عملوندهای عملگرهای < و > تنها از جنس اعداد صحیح هستند. دو عملگر == و != روی تمامی تایپ‌ها تعریف شده‌اند و باید تایپ عملوندهای آنها یکسان باشد. در صورت لیست بودن، باید تعداد اعضای دو لیست و نوع اعضای متناظر آنها برابر باشد. در غیر این صورت باید خطای کامپایل گرفته شود. توجه داشته باشید که هر متغیر می‌تواند با استفاده از عملگرهای == و != با مقدار اولیه تایپ خود نیز مقایسه شود.

لیست عملگرهای مقایسه‌ای در جدول زیر آمده است. مقادیر A و B را همانند قبل در نظر بگیرید.

عملگر	شرکت‌پذیری	توضیح	مثال
==	چپ	تساوی	(A == B) = false
!=	چپ	عدم تساوی	(A != B) = true
>	چپ	کوچکتر	(A < B) = false
<	چپ	بزرگتر	(A > B) = true

۶-۳. عملگرهای منطقی

در این زبان، عملگرهای منطقی تنها روی تایپ bool قابل اعمال است. این عملگرها در جدول زیر آمده است. A را برابر true و B را برابر false در نظر بگیرید.

مثال	توضیح	شرکت پذیری	عملگر
$(A \ \&\& \ B) = \text{false}$	عطف منطقی	چپ	$\&\&$
$(A \ \ B) = \text{true}$	فصل منطقی	چپ	$ $
$(!A) = \text{false}$	نقیض منطقی	راست	$!$

۶-۴. عملگر تخصیص

این عملگر که به صورت $=$ نمایش داده می شود وظیفه ی تخصیص را برعهده دارد. یعنی مقدار عملوند سمت راست را به عملوند سمت چپ اختصاص می دهد. برای لیست مقدار تک تک عناصر لیست سمت راست را به عناصر لیست سمت چپ تخصیص می دهد.

دقت داشته باشید که عملوند سمت چپ باید از نوع lvalue باشد. مفهوم rvalue و lvalue در زبان Sophia مشابه زبان C است. عبارات lvalue عباراتی هستند که به یک مکان در حافظه اشاره می کنند. در مقابل عبارات rvalue به مکان خاصی در حافظه اشاره نمی کنند و صرفاً یک عبارت دارای مقدار هستند. به عنوان مثال یک متغیر یا یک دسترسی به یکی از عناصر لیست یک عبارت lvalue است اما عبارت $5 + 1$ یک عبارت rvalue محسوب می شود. عبارات rvalue تنها در سمت راست عملگر تخصیص قرار می گیرند.

۶-۵. اولویت عملگرها

اولویت	دسته	عملگرها	شرکت پذیری
۱	پرانتز	$()$	چپ به راست
۲	دسترس به اعضاء	$.$	چپ به راست

چپ به راست	[]	دسترسی به عناصر لیست	۳
چپ به راست	++ --	تک عملوندی پسوندی	۴
راست به چپ	-- ++ ! -	تک عملوندی پیشوندی	۵
چپ به راست	% / *	ضرب و تقسیم	۶
چپ به راست	+	جمع و تفریق	۷
چپ به راست	< >	رابطه‌ای	۸
چپ به راست	== !=	مقایسه‌ی تساوی	۹
چپ به راست	&&	عطف منطقی	۱۰
چپ به راست		فصل منطقی	۱۱
راست به چپ	=	تخصیص	۱۲
چپ به راست	,	کاما (ورودی متدها)	۱۳

۷. ساختار تصمیم‌گیری

در زبان Sophia تنها ساختار تصمیم‌گیری، if..else است. ساختار نحوی^۱ آن مشابه زیر است:

^۱ syntax

```

1 if (x[2] > 0) {
2     print("positive");
3 } else if (x[2] < 0) {
4     print("negative");
5 } else {
6     print("zero");
7 }

```

این ساختار بدون else نیز می‌تواند به کار رود.

۸. ساختار تکرار

۸-۱. حلقه‌ی for

ساختار نحوی آن به این صورت است که ۳ عبارت به عنوان مقداردهی، شرط و بروزرسانی می‌گیرد که عبارت مقداردهی و بروزرسانی آن یک گزاره است و عبارت شرط هر عبارتی از تایپ bool می‌تواند باشد. هرکدام از عبارات مقداردهی، شرط و بروزرسانی for می‌توانند خالی باشند. دقت شود که تمامی متغیرها در ابتدای متدها تعریف می‌شوند و متغیری را در ۳ عبارت بالا یا در داخل اسکوپ for تعریف نمی‌کنیم.

```

1 for(initializationStatement;conditionExpression;updateStatement) {
2     // statements inside the body of loop
3 }

```


۸-۲. حلقه‌ی foreach

در ساختار نحوی این حلقه از کلیدواژه in استفاده شده است. سمت چپ این کلیدواژه یک متغیر که قبلاً تعریف شده و سمت راست آن یک عبارت است. این حلقه برای پیمایش اعضای لیست‌هایی که تمامی اعضای آنها از یک جنس باشد تعبیه شده است.

```
1 foreach (identifier in list) {  
2     // statements inside the body of loop  
3 }
```

۸-۳. کلیدواژه break

در هر کدام از حلقه‌های تعریف شده در زبان Sophia می‌توان با استفاده از کلیدواژه break از حلقه خارج شد و دستورات بعد از حلقه را اجرا کرد. بدیهی است که در صورت استفاده از چند حلقه‌ی تودرتو، دستور break روی درونی‌ترین حلقه عمل می‌کند و اجرای حلقه‌های بیرونی ادامه پیدا می‌کند.

۸-۴. کلیدواژه continue

در هر کدام از حلقه‌های تعریف شده در زبان Sophia می‌توان با استفاده از کلیدواژه continue تکرار کنونی حلقه را متوقف و تکرار بعدی را آغاز کرد. بدیهی است که در صورت استفاده از چند حلقه‌ی تودرتو، دستور continue روی درونی‌ترین حلقه عمل می‌کند.

۹. scope ها

۹-۱. scope های موجود در زبان

به طور کلی در زبان Sophia موارد زیر در scope جدیدی قرار دارند:

- خطوط کد داخل یک کلاس
- پارامترها و خطوط کد داخل یک متد
- داخل گزاره‌های تصمیم‌گیری و تکرار
- داخل scope های جدیدی که با {} مشخص می‌شوند

۹-۲. قوانین scope ها

- تعریف کلاس‌ها در بیرونی‌ترین scope است.
- خطوط خالی از کد اجرایی هیچ تاثیری در خروجی و اجرای برنامه ندارد.
- کدهای داخل هر متد در scope آن متد هستند.
- متغیرهایی که داخل یک scope تعریف می‌شوند در scope های بیرون آن دسترس‌پذیر نیستند و صرفاً در scope های درون آن قابل دسترسی هستند.
- امکان تعریف متغیر با نام یکسان در یک scope وجود ندارد اما در scope های درونی آن امکان تعریف مجدد وجود دارد و تا زمان خروج از scope درونی، نزدیکترین تعریف به آن استفاده می‌شود.

۹-۳. قوانین خطوط برنامه

قوانین خطوط این زبان مشابه زبان Java است؛ تمامی دستورات، در انتهای خود یک کاراکتر ; دارند.

۱۰. توابع پیش فرض

در زبان Sophia تنها یک تابع پیش فرض وجود دارد و آن هم print است. این تابع به صورت ضمنی تعریف شده است و می‌تواند یک مقدار int، bool یا string دریافت کند و مقدار آن را در کنسول چاپ کند.



```
1 print("Hello World!");
```

۱۱. نمونه کد

نمونه‌هایی از کدهای نوشته شده در زبان Sophia در ادامه آمده است.

```
1 class Person {
2     name: string;
3     age: int;
4     def Person(name: string, age: int) {
5         this.name = name;
6         this.age = age;
7     }
8     def void showName() {
9         print(this.name);
10        print("\n");
11    }
12 }
13
14 class Employee extends Person {
15     salary: int;
16     def Employee(name: string, age: int, salary: int) {
17         this.name = name;
18         this.age = age;
19         this.salary = salary;
20     }
21     def void showSalary() {
22         print("Salary is:");
23         print(this.salary);
24         print("\n");
25     }
26 }
27
28 class Main {
29     def Main() {
30         e: Employee;
31         p: Person;
32         e = new Employee("Hannah", 21, 3000);
33         e.showName();
34         e.showSalary();
35         p = new Person("Sophia", 30);
36         p.showName();
37     }
38 }
```

```

1 class Cart {
2     orders: list(4 # list(name: string, price: int, quantity: int));
3     def void addToCart(product: list(string, int, int), idx: int) {
4         this.orders[idx] = product;
5     }
6     def int getSum() {
7         total: int;
8         current: int;
9         order: list(name: string, price: int, quantity: int);
10        total = 0;
11        foreach (order in this.orders) {
12            if (order.quantity == 0)
13                break;
14            current = order.price * order.quantity + order.quantity *
15                    100 % 1000 - order.price / 100;
16            total = total + current;
17        }
18        return total;
19    }
20 }
21 class Main{
22     def Main() {
23         cart: Cart;
24         i: int;
25         total: int;
26         order: list(string, int, int);
27         product: list(string, int);
28         productCatalog: list(4 # list(string, int));
29         productCatalog = [{"Doughnut", 5000}, {"Croissant", 4000},
30                            {"Cookies", 2000},
31                            {"Chocolate Cake", 8000}];
32         cart = new Cart();
33         for (i = 0; i < 4; i = i + 1) {
34             product = productCatalog[i];
35             order = [product[0], product[1], i + 1];
36             cart.addToCart(order, i);
37         }
38         total = cart.getSum();
39         print(total);
40         print("\n");
41     }
42 }

```